
==========S_Z_Blocks==========
*                         _______
*         IN   RS=1/FC  _|_      |  Integration
* ___   |\              ___ CS   |  Block (1/s)
*|VIN|__| \_/\  /\  /\___|___|\  |  ___
*|___|  | /   \/  \/   VNEG  | \_|_|OUT|
*       |/ Gain -1       ____| /   |___|
*                       _|_  |/
*   For FC = 1Hz        ///
*   RS = 1 Ohm               CS=1uF/2Pi
*   Xc = 1 Ohm
*
*      F1 = 2*sin(pi*FC/FS)     Integration
*               ___             Block (1/z)
* ___   |\     / _ \            ___
*|VIN|__| \__\/ \   \__________|OUT|
*|___|  | /  /\ /_  /        | |___|
*       |/     \___/  | CLK  |
*      Gain F1  ^   __V____  |
*               |__| Z_DLAY|_|
*                  |_______|

The "1/s" integration block can be done digitally in 
a very similar manner. The digital "1/z" integration block
sums a scaled value of the input signal with the accumulated 
of what has happened in the past.  

*=========Create_Signal==================
VT        VT     0      DC     0   PWL( 0 0 100 100)
BVIN      VIN    0      V      =  1*sin(6.283*V(FIN)*V(VT))
BVCLK     CLK    0      V      =  u(sin(6.283*V(FS)*V(VT)))
BF1       F1     0      V      =  2*sin(3.14*V(FC)/V(FS))

VFIN      FIN    0      DC     1
VFC       FC     0      DC     1
VFS       FS     0      DC     50

*=========Compare_S_to_Z==================
XSblock   VIN    VOUTS   FC          Sblock
XZ_DLAY   VIN    VINZ          CLK   Z_DLAY
XZblock   VINZ   OUTZ    F1    CLK   Zblock

The two integration blocks should be able to behave the same.
The "1/z" block does however need to taking into account the
clock rate to get the frequency scaling right.  

*=========Z_BLOCK===========================================
*
*      F1 = 2*sin(pi*Fc/Fs)
*               ___
* ___   |\ IN1 / _ \            ___
*|VIN|__|1\__\/ \   \__________|OUT|
*|___|  | /  /\ /_  /        | |___|
*       |/     \___/  | CLK  |
*          IN2  ^   __V____  |
*               |__| Z_DLAY|_|
*                  |_______|
*
.SUBCKT   Zblock VINZ   OUTZ   F1   CLK 
 B1       IN1    0      V =    V(VINZ)*V(F1)
 B2       OUTZ   0      V =    V(IN1) + V(IN2)
 XZ_DLAY  OUTZ   IN2    CLK    Z_DLAY
.ENDS     Zblock

The actual digital circuit really only consists of VOUT register
which holds the VOUT value in binary format. At every clock
cycle, a scaled down value of the input gets added to the 
VOUT value.  



*=========Z_DLAY=========================================
*                          _________________
*                          |   ___________  |
*  ___   |\          VBF   |  |  ____     | | ____
* |CLK|__|1\_______________|_/|\_\   \    | |_|   \
* |___|  | / |           VLP  |   |NE \_  |   | PE \_
*        |/  |/\  /\  /\______|___| 2 / | |___| 3  / |
*               \/  \/  _|_      /___/  |     |___/  |
*               RLP     ___                          |
*                       _|_
*                   CLP ///
*
* ___   |\  V1  /   V2       |\  V3  /   V4     |\   ___
*|IN |__|4\____/ ____________|5\____/ __________|6\_|OUT|
*|___|  | / S1  NE  | _|_    | / S2  PE  | _|_  | / |___|
*       |/          | ___    |/          | ___  |/
*        Rshunt1    |  | C1   Rshunt2    |  |C2
*      _/\  /\  /\__| _|_   _/\  /\  /\__| _|_
*    _|_  \/  \/      /// _|_  \/  \/      ///
*    ///                  ///
.SUBCKT   Z_DLAY VIN    OUT    CLK  
 B1       VBF    0      V =    u( v(CLK )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      50n    IC=0
 BNOR2    NE     0      V =    1-u( u(v(VBF )-.5)+u(.5 -v(VLP ) ) -.1)
 BAND3    PE     0      V =      u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
 B4       V1     0      V =    V(VIN)
 S1       V1     V2     NE     0    SW 
 R1       V2     0      100Meg
 C1       V2     0      1u
 B5       V3     0      V =    V(V2)
 S2       V3     V4     PE     0    SW 
 R2       V4     0      100Meg
 C2       V4     0      100u
 B6       OUT    0      V =    V(V4)
.ENDS     Z_DLAY

This is another case where the actual digital circuit is easier
than the subcircuit. This method is being used to avoid convergence 
problems. The actual circuit is just a shift register.

.control
*=========Test_At_Unity_Gain=================
set       pensize = 2
run
plot      vin vouts  vinz outz  title FIN_eq_1_FS_eq_50
*=========Test_At_Twice_Input_Frequency==============
alter     VFIN         dc = 2 
run
let       InFreq     =       mean(FIN)
echo     "InputFrequencyNew  $&InFreq"
plot      vin vouts  vinz outz  title FIN_eq_2_FS_eq_50
*=========Test_At_Half_Input_Frequency==============
alter     VFIN         dc = .5 run
run
let       InFreq     =       mean(FIN)
echo     "InputFrequencyNew  $&InFreq"
plot      vin vouts  vinz outz  title FIN_eq_.5_FS_eq_50
*=========Test_At_Half_Sample_Rate==============
alter     VFS         dc = 25 
run
let       SampleFreq =       mean(FS)
echo     "SampleFrequencyNew $&SampleFreq"
plot      vin vouts  vinz outz title FIN_eq_.5_FS_eq_25
.endc

In this example, the input frequency and  clock frequency are being
adjusted to show how the integration is taking place in effectively the
same way.  



The frequency scaling has been set so at 1Hz, the input and output should
be equal in magnitude. Notice that the input input needs to be sample 
before applying to the digital integrator. 

Changing the input frequency to 2Hz should have the same effect on both
the "1/s" block and the "1/z" block. 

And Changing the input frequency to 0.5Hz should have the same effect.



Halving the sample frequency needs to adjust a scale factor in the
digital integrator "1/Z" block in order to maintain the asme sample 
output voltage level. 

===========Full_Netlist_For_Copy_Paste=======================
S_Z_Blocks
*                         _______
*         IN   RS=1/FC  _|_      |  Integration
* ___   |\              ___ CS   |  Block (1/s)
*|VIN|__| \_/\  /\  /\___|___|\  |  ___
*|___|  | /   \/  \/   VNEG  | \_|_|OUT|
*       |/ Gain -1       ____| /   |___|
*                       _|_  |/
*   For FC = 1Hz        ///
*   RS = 1 Ohm               CS=1uF/2Pi
*   Xc = 1 Ohm
*
*      F1 = 2*sin(pi*Fc/Fs)     Integration
*               ___             Block (1/z)
* ___   |\     / _ \            ___
*|VIN|__| \__\/ \   \__________|OUT|
*|___|  | /  /\ /_  /        | |___|
*       |/     \___/  | CLK  |
*      Gain F1  ^   __V____  |
*               |__| Z_DLAY|_|
*                  |_______|
*=========Create_Signal==================
VT        VT     0      DC     0   PWL( 0 0 100 100)
VFIN      FIN    0      DC     1
VFC       FC     0      DC     1
VFS       FS     0      DC     50
BVIN      VIN    0      V      =  1*sin(6.283*V(FIN)*V(VT))
BVCLK     CLK    0      V      =  u(sin(6.283*V(FS)*V(VT)))
BF1       F1     0      V      =  2*sin(3.14*V(FC)/V(FS))
XSblock   VIN    VOUTS  FC           Sblock
XZ_DLAY   VIN    VINZ   CLK          Z_DLAY
XZblock   VINZ   OUTZ   F1     CLK   Zblock
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
.tran     30u    2      0      30u    UIC
.control
*=========Test_At_Unity_Gain=================
set       pensize = 2
run
plot      vin vouts  vinz outz  title FIN_eq_1_FS_eq_50
let       InFreq     =       mean(FIN)
let       UnityFreq  =       mean(FC)
let       SampleFreq =       mean(FS)
let       F1_Scale   =       mean(F1)
echo     "InputFrequency     $&InFreq"
echo     "UnityFrequency     $&UnityFreq"
echo     "SampleFrequency    $&SampleFreq"
echo     "F1_Scale           $&F1_Scale"
*=========Test_At_Twice_Input_Frequency==============
alter     VFIN         dc = 2 
run
let       InFreq     =       mean(FIN)
echo     "InputFrequencyNew  $&InFreq"
plot      vin vouts  vinz outz  title FIN_eq_2_FS_eq_50
*=========Test_At_Half_Input_Frequency==============
alter     VFIN         dc = .5 run



run
let       InFreq     =       mean(FIN)
echo     "InputFrequencyNew  $&InFreq"
plot      vin vouts  vinz outz  title FIN_eq_.5_FS_eq_50
*=========Test_At_Half_Sample_Rate==============
alter     VFS         dc = 25 
run
let       SampleFreq =       mean(FS)
echo     "SampleFrequencyNew $&SampleFreq"
plot      vin vouts  vinz outz title FIN_eq_.5_FS_eq_25
.endc
*=========Switch_Model===========================================
.MODEL    SW      SW(    VT=.5 VH=.1  RON=100m ROFF=100MEG)
*=========S_BLOCK===========================================
*                         _______
*         IN   RS=1/FC  _|_      |
* ___   |\              ___      |
*|VIN|__| \_/\  /\  /\___|___|\  |  ___
*|___|  | /   \/  \/   VNEG  | \_|_|OUT|
*       |/               ____| /   |___|
*                       _|_  |/
*   For FC = 1Hz        ///
*   RS = 1 Ohm               CS=1uF/2Pi
*   Xc = 1 Ohm
.SUBCKT   Sblock VIN    OUT    FC  
 Bbuf     IN     0      V =    -V(VIN)
 BRS      IN     VNEG   I =    (V(IN)-V(VNEG))/V(FC)
 Cs       VNEG   OUT    .159      
 BSOUT    OUT    0      V =    -V(VNEG)*1000
.ENDS     Sblock
*=========Z_BLOCK===========================================
*
*      F1 = 2*sin(pi*Fc/Fs)
*               ___
* ___   |\ IN1 / _ \            ___
*|VIN|__|1\__\/ \  2\__________|OUT|
*|___|  | /  /\ /_  /        | |___|
*       |/     \___/  | CLK  |
*          IN2  ^   __V____  |
*               |__| Z_DLAY|_|
*                  |_______|
*
.SUBCKT   Zblock VINZ   OUTZ   F1   CLK 
 B1       IN1    0      V =    V(VINZ)*V(F1)
 B2       OUTZ   0      V =    V(IN1) + V(IN2)
 XZ_DLAY  OUTZ   IN2    CLK    Z_DLAY
.ENDS     Zblock
*=========Z_DLAY=========================================
*                          _________________
*                          |   ___________  |
*  ___   |\          VBF   |  |  ____     | | ____
* |CLK|__|1\_______________|_/|\_\   \    | |_|   \
* |___|  | / |           VLP  |   |NE \_  |   | PE \_
*        |/  |/\  /\  /\______|___| 2 / | |___| 3  / |
*               \/  \/  _|_      /___/  |     |___/  |
*               RLP     ___                          |
*                       _|_
*                   CLP ///
*
* ___   |\  V1  /   V2       |\  V3  /   V4     |\   ___
*|IN |__|4\____/ ____________|5\____/ __________|6\_|OUT|
*|___|  | / S1  NE  | _|_    | / S2  PE  | _|_  | / |___|
*       |/          | ___    |/          | ___  |/
*        Rshunt1    |  | C1   Rshunt2    |  |C2
*      _/\  /\  /\__| _|_   _/\  /\  /\__| _|_
*    _|_  \/  \/      /// _|_  \/  \/      ///
*    ///                  ///
.SUBCKT   Z_DLAY VIN    OUT    CLK  
 B1       VBF    0      V =    u( v(CLK )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      50n    IC=0
 BNOR2    NE     0      V =    1-u( u(v(VBF )-.5)+u(.5 -v(VLP ) ) -.1)
 BAND3    PE     0      V =      u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
 B4       V1     0      V =    V(VIN)
 S1       V1     V2     NE     0    SW 
 R1       V2     0      100Meg
 C1       V2     0      1u
 B5       V3     0      V =    V(V2)
 S2       V3     V4     PE     0    SW 
 R2       V4     0      100Meg
 C2       V4     0      100u
 B6       OUT    0      V =    V(V4)
.ENDS     Z_DLAY
.end

